Interpreting Linear Models (Especially Interactions)

- Today's Class:
  - Representing effects of categorical predictors
  - Decomposing interactions among continuous predictors
  - (see example for interactions among categorical predictors)

The Two Sides of a Model

$$y_i = \boxed{\beta_0 + \beta_1 X_i + \beta_2 Z_i + \beta_3 X_i Z_i} + \boxed{e_i}$$

Our focus today

- **Model for the Means (Predicted Values):**
  - Each person's expected (predicted) outcome is a weighted linear function of his/her values on X and Z (and here, their interaction), each measured once per person (i.e., this is a between-person model)
  - **Estimated parameters are called fixed effects** (here, β_0, β_1, β_2, and β_3)
  - The number of fixed effects will show up in formulas as ***k*** (so ***k* = *4*** here)

- **Model for the Variance ("Piles" of Variance):**
  - $e_i \sim N(0, \sigma_e^2)$→ ONE residual (unexplained) deviation
  - e_i has a mean of 0 with some estimated constant variance σ_e^2, is normally distributed, is unrelated to X and Z, and is unrelated across people (across all observations, just people here)
  - **Estimated parameter is the residual variance only** (in above model)

Representing the Effects of Predictors

- From now on, we will think carefully about exactly **how** the **predictor variables** are entered into the **model for the means** (i.e., by which a predicted outcome is created for each person)

- Why don't people always care? Because the scale of predictors:
  - Does NOT affect the amount of outcome variance accounted for (R^2)
  - Does NOT affect the outcomes values predicted by the model for the means (so long as the same predictor fixed effects are included)

- Why should this matter to us?
  - ***Because the Intercept = expected outcome value when X = 0***
  - Can end up with nonsense values for intercept if X = 0 isn't in the data, so we need to change the scale of the predictors to include 0
  - Scaling becomes more important once interactions are included or once random intercepts are included (i.e., variability around fixed intercept)

Adjusting the Scale of Predictors

- For **continuous** (quantitative) predictors, **<u>we</u>** will make the intercept interpretable by **centering**:
  - **Centering** = subtract a constant from each person's variable value so that **the 0 value** falls within the range of the new centered predictor variable
  - Typical → Center around predictor's mean: $Centered\ X_1 = X_1 - \overline{X_1}$
    - Intercept is then expected outcome for "average X_1 person"
  - Better → Center around meaningful constant *C*: $Centered\ X_1 = X_1 - C$
    - Intercept is then expected outcome for person with that constant (even 0 may be ok)

- For **categorical** (grouping) predictors, **<u>either we or the program</u>** will make the intercept interpretable by **creating a reference group**:
  - **Reference group** is given a 0 value on all predictor variables created from the original grouping variable, such that the intercept is the expected outcome for that reference group specifically
  - Accomplished via "dummy coding" (aka, "reference group coding")
    → Two-group example using *Gender*: 0 = Men, 1 = Women
    (or 0 = Women, 1 = Men)

Adjusting the Scale of Predictors

- <u>For more than two groups, need: ***dummy codes = #groups − 1***</u>
  - "Treatgroup" variable: Control=0, Treat1=1, Treat2=2, Treat3=3
  - Variables: *d1*= 0, 1, 0, 0 → difference between Control and T1
    d2= 0, 0, 1, 0 → difference between Control and T2
    d3= 0, 0, 0, 1 → difference between Control and T3

SAS CLASS statement can do this for you ☺

- <u>Potential pit-falls:</u>
  - All predictors for the effect of group (e.g., d1, d2, d3) MUST be in the model at the same time for these specific interpretations to be correct!
  - Model parameters resulting from these dummy codes will not *directly* tell you about differences among non-reference groups (but they can)
- <u>Other examples of things people do to categorical predictors:</u>
  - "Contrast/effect coding"→ *Gender*: −0.5 = Men, 0.5 = Women
  - Test other contrasts among multiple groups → four-group example: *contrast1*= −1, 0.33, 0.33, 0.34 → Control vs. Any Treatment?

Categorical Predictors: Manual Coding

- Model: $y_i = \beta_0 + \beta_1 d1_i + \beta_2 d2_i + \beta_3 d3_i + e_i$
  - "Treatgroup" variable: Control=0, Treat1=1, Treat2=2, Treat3=3
  - New variables to be created for the model:
    d1= 0, 1, 0, 0 → difference between Control and T1
    d2= 0, 0, 1, 0 → difference between Control and T2
    d3= 0, 0, 0, 1 → difference between Control and T3
- How does the model give us **all possible group differences**? By determining each group's mean, and then the difference…

Control Mean (Reference)	Treatment 1 Mean	Treatment 2 Mean	Treatment 3 Mean
$\boldsymbol{\beta_0}$	$\boldsymbol{\beta_0 + \beta_1 d1_i}$	$\boldsymbol{\beta_0 + \beta_2 d2_i}$	$\boldsymbol{\beta_0 + \beta_3 d3_i}$

- The model for the 4 groups directly provides 3 differences (control vs. each treatment), and indirectly provides another 3 differences (differences between treatments)

Group Differences from Dummy Codes

- Model: $y_i = \beta_0 + \beta_1 d1_i + \beta_2 d2_i + \beta_3 d3_i + e_i$

Control Mean (Reference)	Treatment 1 Mean	Treatment 2 Mean	Treatment 3 Mean
$\boldsymbol{\beta_0}$	$\boldsymbol{\beta_0+\beta_1 d1_i}$	$\boldsymbol{\beta_0+\beta_2 d2_i}$	$\boldsymbol{\beta_0+\beta_3 d3_i}$

	Alt Group		Ref Group	Difference
• Control vs. T1 =	$(\beta_0+\beta_1)$	$-$	(β_0)	$= \beta_1$
• Control vs. T2 =	$(\beta_0+\beta_2)$	$-$	(β_0)	$= \beta_2$
• Control vs. T3 =	$(\beta_0+\beta_3)$	$-$	(β_0)	$= \beta_3$
• T1 vs. T2 =	$(\beta_0+\beta_2)$	$-$	$(\beta_0+\beta_1)$	$= \beta_2 - \beta_1$
• T1 vs. T3 =	$(\beta_0+\beta_3)$	$-$	$(\beta_0+\beta_1)$	$= \beta_3 - \beta_1$
• T2 vs. T3 =	$(\beta_0+\beta_3)$	$-$	$(\beta_0+\beta_2)$	$= \beta_3 - \beta_2$

ESTIMATEs when using dummy codes

	Alt Group	Ref Group	Difference
• Control vs. T1 =	$(\beta_0+\beta_1)$	$-\ (\beta_0)$	$= \beta_1$
• Control vs. T2 =	$(\beta_0+\beta_2)$	$-\ (\beta_0)$	$= \beta_2$
• Control vs. T3 =	$(\beta_0+\beta_3)$	$-\ (\beta_0)$	$= \beta_3$
• T1 vs. T2 =	$(\beta_0+\beta_2)$	$-\ (\beta_0+\beta_1)$	$= \beta_2 - \beta_1$
• T1 vs. T3 =	$(\beta_0+\beta_3)$	$-\ (\beta_0+\beta_1)$	$= \beta_3 - \beta_1$
• T2 vs. T3 =	$(\beta_0+\beta_3)$	$-\ (\beta_0+\beta_2)$	$= \beta_3 - \beta_2$

Note the order of the equations: the reference group mean *is subtracted from* the alternative group mean.

In ESTIMATE statements, the variables refer to their betas; the numbers refer to the operations of their betas.

```
TITLE "Manual Contrasts for 4-Group Diffs";
PROC MIXED DATA=dataname ITDETAILS METHOD=ML;
MODEL y = d1 d2 d3 / SOLUTION;
ESTIMATE "Control Mean" intercept 1 d1 0 d2 0 d3 0;
ESTIMATE "T1 Mean"      intercept 1 d1 1 d2 0 d3 0;
ESTIMATE "T2 Mean"      intercept 1 d1 0 d2 1 d3 0;
ESTIMATE "T3 Mean"      intercept 1 d1 0 d2 0 d3 1;
ESTIMATE "Control vs. T1"     d1  1 d2  0 d3 0;
ESTIMATE "Control vs. T2"     d1  0 d2  1 d3 0;
ESTIMATE "Control vs. T3"     d1  0 d2  0 d3 1;
ESTIMATE "T1 vs. T2"          d1 -1 d2  1 d3 0;
ESTIMATE "T1 vs. T3"          d1 -1 d2  0 d3 1;
ESTIMATE "T2 vs. T3"          d1  0 d2 -1 d3 1;
RUN;
```

Intercepts are used <u>only</u> in predicted values.

Positive values indicate addition; negative values indicate subtraction.

Using the CLASS statement instead

- If you let SAS do the dummy coding instead via CLASS, then the **highest/last group is the reference**
- **Manual** model: $y_i = \beta_0 + \beta_1 d1_i + \beta_2 d2_i + \beta_3 d3_i + e_i$
  - "Treatgroup" variable: Control=0, Treat1=1, Treat2=2, Treat3=3
  - New variables you created for the model:
    - *d1*= 0, 1, 0, 0 → difference between Control and T1
    - *d2*= 0, 0, 1, 0 → difference between Control and T2
    - *d3*= 0, 0, 0, 1 → difference between Control and T3
  - When including d1, d2, and d3, SAS doesn't understand they are part of one 4-group variable, and **so does not provide omnibus (df=3) F-tests**

- **CLASS** model: $y_i = \beta_0 + \beta_1 g0_i + \beta_2 g1_i + \beta_3 g2_i + e_i$
  - New variables created by using CLASS:
    - *g0*= 1, 0, 0, 0 → difference between T3 and Control
    - *g1*= 0, 1, 0, 0 → difference between T3 and T1
    - *g2*= 0, 0, 1, 0 → difference between T3 and T2
  - If SAS does the coding, it will provide 4-group (df=3) omnibus F-tests (and compute all cell means and differences using **LSMEANS**)

Using the CLASS statement instead

- CLASS model: $y_i = \beta_0 + \beta_1 g0_i + \beta_2 g1_i + \beta_3 g2_i + e_i$
  - New variables created by using CLASS:
    - $g0$= 1, 0, 0, 0 → difference between T3 and Control
    - $g1$= 0, 1, 0, 0 → difference between T3 and T1
    - $g2$= 0, 0, 1, 0 → difference between T3 and T2

```
TITLE "CLASS Contrasts for 4-Group Differences";
PROC MIXED DATA=dataname ITDETAILS METHOD=ML;
CLASS treatgroup;
MODEL y = treatgroup / SOLUTION;
LSMEANS treatgroup / DIFF=ALL;

ESTIMATE "Control Mean"        intercept 1 treatgroup 1 0 0 0;
ESTIMATE "T1 Mean"             intercept 1 treatgroup 0 1 0 0;
ESTIMATE "T2 Mean"             intercept 1 treatgroup 0 0 1 0;
ESTIMATE "T3 Mean"             intercept 1 treatgroup 0 0 0 1;
ESTIMATE "Control vs. T1"      treatgroup  -1  1  0  0;
ESTIMATE "Control vs. T2"      treatgroup  -1  0  1  0;
ESTIMATE "Control vs. T3"      treatgroup  -1  0  1  0;
ESTIMATE "T1 vs. T2"           treatgroup   0 -1  1  0;
ESTIMATE "T1 vs. T3"           treatgroup   0 -1  0  1;
ESTIMATE "T2 vs. T3"           treatgroup   0  0 -1  1;
RUN;
```

Note that treatgroup is the only predictor.

This LSMEANS line provides the same information as all statements below!

Treatgroup has 4 possible levels, so 4 values must be given in ESTIMATEs.

To CLASS or not to CLASS?

- Letting SAS create dummy codes for categorical predictors (instead of creating manual dummy codes) does the following:
  - Allows use of LSMEANS (for cell means and differences)
  - Provides omnibus (multiple df) group F-tests
  - Marginalizes the group effect across interacting predictors
    → omnibus F-tests represent marginal main effects (instead of simple)
  - e.g., MODEL y = Treatgroup Gender Treatgroup*Gender
    (in which Treatgroup is always on CLASS statement)

Type 3 Tests of Fixed Effects	**Interpretation if using dummy code for Gender**	**Interpretation if CLASS statement for Gender**
Gender	Marginal gender diff	Marginal gender diff
Treatgroup	Group diff if gender=0	Marginal group diff
Treatgroup*Gender	Interaction	Interaction

Continuous Predictors

- For **continuous** (quantitative) predictors, **we** (not SAS) will make the intercept interpretable by **centering**
  - **Centering** = subtract a constant (e.g., sample mean, other meaningful reference value) from each person's variable value so that **the 0 value** falls within the range of the new centered predictor variable
  - **Continuous predictors do not go on the CLASS statement**
  - Predicted group means **at** specific levels of continuous predictors can be found using LSMEANS (e.g., if X1 SD=5, means at ±1 SD):
    - ```
      CLASS treatgroup;
      MODEL y = treatgroup x1 treatgroup*x1 / SOLUTION;
      LSMEANS treatgroup / AT (x1)=(-5) DIFF=ALL;
      LSMEANS treatgroup / AT (x1)=( 0) DIFF=ALL;
      LSMEANS treatgroup / AT (x1)=( 5) DIFF=ALL;
      ```
  - Continuous predictors cannot be used on LSMEANS otherwise
      ```

Interactions: $y_i = \beta_0 + \beta_1 X_i + \beta_2 Z_i + \beta_3 X_i Z_i + e_i$

- **Interaction = Moderation:** the effect of a predictor depends on the value of the interacting predictor
  - Either predictor can be "the moderator" (interpretive distinction only)
- Interactions can always be evaluated for any combination of categorical and continuous predictors, although traditionally…
  - In "ANOVA": By default, all possible interactions are estimated
    - Software does this for you; oddly enough, nonsignificant interactions usually still are kept in the model (even if only significant interactions are interpreted)
  - In "ANCOVA": Continuous predictors ("covariates") do not get to be part of interaction terms → "homogeneity of regression assumption"
    - There is no reason to assume this – it is a testable hypothesis!
  - In "Regression": No default – effects of predictors are as you specify
    - Requires most thought, but gets annoying because in regression programs you usually have to manually create the interaction as an observed variable:
    - e.g., XZinteraction = centeredX * centeredZ

Interaction variables are created for you in SAS PROC GLM, MIXED, and GLIMMIX ☺

Main Effects of Predictors within Interactions

- Main effects of predictors within interactions should remain in the model regardless of whether or not they are significant
  - An interaction is an over-additive (enhancing) or under-additive (dampening) effect, so *what it is additive to* must be included

- The role of a two-way interaction is to adjust its main effects…
- However, the idea of a "main effect" no longer applies… each main effect is ***conditional*** on the interacting predictor = 0

- e.g., Model of Y = W, X, Z, X*Z:
  - The effect of W is still a "main effect" because it is not part of an interaction
  - The effect of X is now the conditional main effect of X *specifically when Z=0*
  - The effect of Z is now the conditional main effect of Z *specifically when X=0*

- The trick is keeping track of what 0 means for every interacting predictor, which depends on the way each predictor is being represented, as determined by you, or by the software without you!

Interactions: Why 0 Matters

- Y = Student achievement (GPA as percentage out of 100)
  X = Parent attitudes about education (measured on 1-5 scale)
  Z = Father's education level (measured in years of education)
- $GPA_i = \beta_0 + (\beta_1 * Att_i) + (\beta_2 * Ed_i) + (\beta_3 * Att_i * Ed_i) + e_i$
  $GPA_i = 30 + (1 * Att_i) + (2 * Ed_i) + (0.5 * Att_i * Ed_i) + e_i$
- **Interpret β_0:**
- **Interpret β_1:**
- **Interpret β_2:**
- **Interpret β_3: Attitude as Moderator:**

  Education as Moderator:

- **Predicted GPA** for **attitude of 3** and **Ed of 12**?

  $75 = 30 + 1*(3) + 2*(12) + 0.5*(3)*(12)$

Model-Implied Simple Main Effects

- **Original:** $GPA_i = \beta_0 + (\beta_1 * Att_i) + (\beta_2 * Ed_i) + (\beta_3 * Att_i * Ed_i) + e_i$
  $GPA_i = 30 + (1 * Att_i) + (2 * Ed_i) + (0.5 * Att_i * Ed_i) + e_i$
- Given any values of the predictor variables, the model equation provides predictions for:
  - Value of outcome (model-implied intercept for non-zero predictor values)
  - Any conditional (simple) main effects implied by an interaction term
  - **Simple Main Effect = what it is + what *modifies* it**

- Step 1: **Identify** all terms in model involving the predictor of interest
  - e.g., Effect of Attitudes comes from: $\beta_1 * Att_i + \beta_3 * Att_i * Ed_i$
- Step 2: **Factor out** common predictor variable
  - Start with $[\beta_1 * Att_i + \beta_3 * Att_i * Ed_i] \rightarrow [Att_i (\beta_1 + \beta_3 * Ed_i)] \rightarrow Att_i$ **(new** β_1**)**
  - Value given by ( ) is then the model-implied coefficient for the predictor
- Step 3: **ESTIMATEs** calculate model-implied simple effect and SE
  - Let's try it for **a new reference point of attitude = 3 and education = 12**

Model-Implied Simple Main Effects

- Old Equation using uncentered predictors:
  $GPA_i = \beta_0 + (\beta_1 * Att_i) + (\beta_2 * Ed_i) + (\beta_3 * Att_i * Ed_i) + e_i$
  $GPA_i = 30 + (1 * Att_i) + (2 * Ed_i) + (0.5 * Att_i * Ed_i) + e_i$
- New equation using centered predictors ($Att_i - 3$ and $Ed_i - 12$):
  $GPA_i = \_\_ + \_\_\_ * (Att_i - 3) + \_\_\_ * (Ed_i - 12) + \_\_\_ * (Att_i - 3) * (Ed_i - 12) + e_i$

- **Intercept: expected value of GPA when $Att_i = 3$ and $Ed_i = 12$**
  $\beta_0 = 75$
- **Simple main effect of Att if $Ed_i = 12$**
  $\beta_1 * Att_i + \beta_3 * Att_i * Ed_i \rightarrow Att_i(\beta_1 + \beta_3 * Ed_i) \rightarrow Att_i(1 + 0.5 * 12)$
- **Simple main effect of Ed if $Att_i = 3$**
  $\beta_2 * Ed_i + \beta_3 * Att_i * Ed_i \rightarrow Ed_i(\beta_2 + \beta_3 * Att_i) \rightarrow Ed_i(2 + 0.5 * 3)$
- **Two-way interaction of Att and Ed:**
  $(0.5 * Att_i * Ed_i)$

Model-Implied Simple Main Effects

- Old Equation using uncentered predictors:
  $GPA_i = \beta_0 + (\beta_1 * Att_i) + (\beta_2 * Ed_i) + (\beta_3 * Att_i * Ed_i) + e_i$
  $GPA_i = 30 + (1 * Att_i) + (2 * Ed_i) + (0.5 * Att_i * Ed_i) + e_i$
- **Intercept: expected value of GPA when $Att_i = 3$ and $Ed_i = 12$**
- **Simple main effect of Att if $Ed_i = 12$ → $Att_i(\beta_1 + \beta_3 * Ed_i)$**
- **Simple main effect of Ed if $Att_i = 3$ → $Ed_i(\beta_2 + \beta_3 * Att_i)$**

```
TITLE "Calculating Model-Implied Parameters";
PROC MIXED DATA=dataname ITDETAILS METHOD=ML;
MODEL y = att ed att*ed / SOLUTION;
ESTIMATE "GPA if Att=3, Ed=12"    intercept 1 att 3 ed 12 att*ed 36;
ESTIMATE "Effect of Att if Ed=12" att 1 att*ed 12;
ESTIMATE "Effect of Ed if Att=3"   ed 1 att*ed 3;
RUN;
```

In ESTIMATE statements, the variables refer to their betas; the numbers refer to the operations of their betas.

These estimates would be given directly by the model parameters instead if you re-centered the predictors as: Att-3, Ed-12.

More Generally…

- Can decompose a **2-way interaction** by testing the simple effect of X at different levels of Z (and vice-versa)
  - Use ESTIMATEs to request simple effects at any point of the interacting predictor
  - Re-centering the interacting predictor at those points will also work

- More general rules, given a **3-way interaction**:
  - *Simple (main) effects move the intercept*
    - 1 possible interpretation for each simple main effect
    - Each simple effect is conditional on other two variables = 0
  - *The 2-way interactions (3 of them in a 3-way model) move the simple effects*
    - 2 possible interpretations for each 2-way interaction
    - Each 2-way interaction is conditional on third variable = 0
  - *The 3-way interaction moves each of the 2-way interactions*
    - 3 possible interpretations of the 3-way interaction
    - Is highest-order term in model, so is unconditional (applies always)

Practice with 3-Way Interactions

- Intercept = 5, Effect of X = 1.0, Effect of Z = 0.50, Effect of W = 0.20
- **X*Z = .10 (applies specifically when W is 0)**
  - #1: for every 1-unit ΔX,
  - #2: for every 1-unit ΔZ,
- **X*W = .01 (applies specifically when Z is 0)**
  - #1: for every 1-unit ΔX,
  - #2: for every 1-unit ΔW,
- **Z*W = .05 (applies specifically when X is 0)**
  - #1: for every 1-unit ΔZ,
  - #2: for every 1-unit ΔW,
- **X*Z*W = .001 (unconditional because is highest order)**
  - #1: for every 1-unit ΔX,
  - #2: for every 1-unit ΔZ,
  - #3: for every 1-unit ΔW,

Practice with 3-Way Interactions

- Model: $y_i = \beta_0 + \beta_1 X_i + \beta_2 Z_i + \beta_3 W_i + \beta_4 X_i W_i$
  $+\beta_5 X_i Z_i + \beta_6 Z_i W_i + \beta_7 X_i Z_i W_i + e_i$
- Calculate simple main effects:
  - For X →
  - For Z →
  - For W →

- Calculate simple 2-way interactions:
  - For X*Z →
  - For X*W →
  - For Z*W →

Interpreting Interactions: Summary

- Interactions represent "moderation" – the idea that the effect of one variable depends upon the level of other(s)

- <u>The main effects WILL CHANGE in once an interaction with them is added, because they now mean different things:</u>
  - Main effect → Simple effect specifically when interacting predictor = 0
  - Best to have 0 as a meaningful predictor value for that reason

- <u>Conditional rules of parameter interpretation:</u>
  - Intercepts are conditional on (i.e., get moved by) main effects
  - Main effects are conditional on two-ways (become 'simple effects')
  - Two-ways are conditional on three-ways… And so forth
  - Highest-order term is unconditional – same regardless of centering